
Sub MsgBoxDemo()
    MsgBox "Klicken Sie auf OK, um mit dem Drucken zu beginnen."
    Sheets("Results").PrintOut
End Sub





Sub GetAnswer()
    Dim Ans As Integer
    Ans = MsgBox("Jetzt drucken?", vbYesNo)
    Select Case Ans
        Case vbYes
            ActiveSheet.PrintOut
        Case vbNo
            MsgBox "Drucken abgebrochen"
    End Select
End Sub





Sub GetAnswer2()
    If MsgBox("Jetzt drucken?", vbYesNo) = vbYes Then
'       ...[Code, wenn auf Ja geklickt wurde]...
    Else
'       ...[Code, wenn nicht auf  Ja geklickt wurde]...
    End If
End Sub





Sub GetAnswer3()
    Dim Config As Integer
    Dim Ans As Integer
    Config = vbYesNo + vbQuestion + vbDefaultButton2
    Ans = MsgBox("Monatsbericht verarbeiten?", Config)
    If Ans = vbYes Then RunReport
End Sub





Sub GetAnswer4()
    Dim Msg As String, Title As String
    Dim Config As Integer, Ans As Integer
    Msg = "Möchten Sie den Monatsbericht verarbeiten?"
    Msg = Msg & vbNewLine & vbNewLine
    Msg = Msg & "Die Verarbeitung des Monatsberichts "
    Msg = Msg & "dauert ca. 15 Minuten. Es wird ein "
    Msg = Msg & "30-seitiger Bericht für alle "
    Msg = Msg & "Verkaufsstellen für den aktuellen "
    Msg = Msg & "Monat erstellt."
    Title = "XYZ Marketing Company"
    Config = vbYesNo + vbQuestion
    Ans = MsgBox(Msg, Config, Title)
    If Ans = vbYes Then RunReport
End Sub





Sub GetName()
    Dim TheName As String
    TheName = InputBox("Wie heißen Sie?", _
       "Greetings", Application.UserName)
End Sub





Sub AddSheets()
    Dim Prompt As String
    Dim Caption As String
    Dim DefValue As Integer
    Dim NumSheets As String

    Prompt = "Wie viele Blätter wollen Sie hinzufügen?"
    Caption = "Sagen Sie es mir..."
    DefValue = 1
    NumSheets = InputBox(Prompt, Caption, DefValue)

    If NumSheets = "" Then Exit Sub 'Canceled
    If IsNumeric(NumSheets) Then
        If NumSheets > 0 Then Sheets.Add Count:=NumSheets
    Else
        MsgBox "Ungültige Zahl"
    End If
End Sub





Sub GetRange()
    Dim Rng As Range
    On Error Resume Next
    Set Rng = Application.InputBox _
      (prompt:="Geben Sie einen Bereich an:", Type:=8)
    If Rng Is Nothing Then Exit Sub
    MsgBox "Ihr ausgewählter Bereich ist " & Rng.Address
End Sub





Sub GetImportFileName ()
    Dim Finfo As String
    Dim FilterIndex As Integer
    Dim Title As String
    Dim FileName As Variant

'   Einrichtung der Dateifilterliste
    FInfo = "Textdateien (*.txt),*.txt," & _
            "Lotus-Dateien (*.prn),*.prn," & _
            "Kommagetrennte Dateien (*.csv),*.csv," & _
            "ASCII-Dateien (*.asc),*.asc," & _
            "All Files (*.*),*.*"

'   Standardmäßig *.* anzeigen
    FilterIndex = 5

'   Titel für das Dialogfeld
    Title = "Eine zu importierende Datei auswählen"

'   Den Dateinamen ermitteln
    FileName = Application.GetOpenFilename (FInfo, _
       FilterIndex, Title)

'   Rückgabeinformation aus dem Dialogfeld verarbeiten
    If FileName = False Then
        MsgBox "Es wurde keine Datei ausgewählt."
    Else
        MsgBox "Ihre Auswahl:" & FileName
    End If
End Sub





Sub GetAFolder()
   With Application.FileDialog(msoFileDialogFolderPicker)
     .InitialFileName = Application.DefaultFilePath & "\"
     .Title = "Wählen Sie bitte einen Speicherort für die Sicherung aus"
     .Show
     If .SelectedItems.Count = 0 Then
        MsgBox "Abgebrochen"
     Else
        MsgBox .SelectedItems(1)
      End If
   End With
End Sub
