
Private Sub Worksheet_Change(ByVal Target As Range)
    If Target.Column = 1 Then
        Target.Offset(0, 1) = Now
    End If
End Sub




Private Sub Workbook_Open()
    Dim Msg As String
    If Weekday(Now) = 6 Then
        Msg = "Heute ist Freitag. Vergessen Sie nicht, "
        Msg = Msg & "den TPS-Bericht abzusenden!"
        MsgBox Msg
    End If
End Sub




Private Sub Workbook_Open()
    Msg = "Dies ist die obercoole Arbeitsmappe von Frank!"
    MsgBox Msg
End Sub




Private Sub Workbook_Open()
    Dim Cnt As Long
    Cnt = GetSetting("MyApp", "Settings", "Open", 0)
    Cnt = Cnt + 1
    SaveSetting "MyApp", "Settings", "Open", Cnt
    MsgBox "Diese Arbeitsmappe wurde " & Cnt & " Mal geöffnet."
End Sub




Private Sub Workbook_BeforeClose(Cancel As Boolean)
    Dim Msg As String
    Dim Ans As Integer
    Dim FName As String
    Msg = "Möchten Sie eine Sicherungskopie für diese Datei anlegen?"
    Ans = MsgBox(Msg, vbYesNo)
    If Ans = vbYes Then
        FName = "F:\BACKUP\" & ThisWorkbook.Name
        ThisWorkbook.SaveCopyAs FName
    End If
End Sub




Private Sub Workbook_BeforeSave(ByVal SaveAsUI _
   As Boolean, Cancel As Boolean)
   Dim Counter As Range
   Set Counter = Sheets("Tabelle1").Range("A1")
   Counter.Value = Counter.Value + 1
End Sub





Private Sub Workbook_BeforeSave(ByVal SaveAsUI _
   As Boolean, Cancel As Boolean)
    If SaveAsUI Then
        MsgBox "Für diese Arbeitsmappe kann keine Kopie angelegt werden!"
        Cancel = True
    End If
End Sub





Private Sub Worksheet_Activate()
    MsgBox "Sie haben aktiviert: " & ActiveSheet.Name
End Sub





Private Sub Worksheet_Activate()
    Range("A1").Activate
End Sub





Private Sub Worksheet_Deactivate()
    MsgBox "Sie müssen in Tabelle1 bleiben"
    Sheets("Tabelle1").Activate
End Sub





Private Sub Workbook_SheetActivate(ByVal Sh As Object)
MsgBox Sh.Name
End Sub





Dim OldSheet As Object

Private Sub Workbook_SheetDeactivate(ByVal Sh As Object)
    Set OldSheet = Sh
End Sub

Private Sub Workbook_SheetActivate(ByVal Sh As Object)
    Dim Msg As String
    If TypeName(Sh) = "Chart" Then
        Msg = "Dieses Diagramm enthält "
        Msg = Msg & ActiveChart.SeriesCollection(1).Points.Count
        Msg = Msg & " Datenpunkte." & vbNewLine
        Msg = Msg & "Mit OK gelangen Sie zurück zu " & OldSheet.Name
        MsgBox Msg
        OldSheet.Activate
    End If
End Sub





Private Sub Workbook_Activate()
    ActiveWindow.WindowState = xlMaximized
End Sub





Private Sub Workbook_Deactivate()
    ThisWorkbook.Windows(1).RangeSelection.Copy
End Sub





Private Sub Worksheet_BeforeDoubleClick _
   (ByVal Target As Excel.Range, Cancel As Boolean)
    Target.Font.Bold = Not Target.Font.Bold
    Cancel = True
End Sub





Private Sub Worksheet_BeforeRightClick _
   (ByVal Target As Excel.Range, Cancel As Boolean)
    If IsNumeric(Target) And Not IsEmpty(Target) Then
        Application.CommandBars.ExecuteMso ("NumberFormatsDialog")
        Cancel = True
    End If
End Sub





Private Sub Worksheet_Change(ByVal Target As Range)
    If Target.Address = "$A$1" Then
        If Not IsNumeric(Target) Then
            MsgBox "Geben Sie eine Zahl in Zelle A1 ein."
            Range("A1").ClearContents
            Range("A1").Activate
        End If
    End If
End Sub





Private Sub Worksheet_Change(ByVal Target As Range)
    Dim VT As Long
    'Haben alle Zellen im Überprüfungsbereich
    'noch eine Überprüfung?
    On Error Resume Next
    VT = Range("InputRange").Validation.Type
    If Err.Number <> 0 Then
        Application.Undo
        MsgBox "Ihre letzte Operation wurde abgebrochen." & _
        "Sie hätte Regeln der Datenüberprüfung gelöscht.", vbCritical
    End If
End Sub





Sub SetAlarm()
    Application.OnTime 0.625, "DisplayAlarm"
End Sub

Sub DisplayAlarm()
    Application.Speech.Speak ("He, aufwachen!")
    MsgBox " Zeit für die Mittagspause!"
End Sub





Application.OnTime TimeValue("3:00:00 pm"), "DisplayAlarm"





Application.OnTime Now + TimeValue("00:20:00"), "DisplayAlarm"





Application.OnTime DateValue("12/31/2013 5:00 pm"), "DisplayAlarm"





Dim NextTick As Date

Sub UpdateClock()
'   Aktualisiert Zelle A1 mit der aktuellen Zeit
    ThisWorkbook.Sheets(1).Range("A1") = Time
'   Legt das nächste Ereignis für fünf Sekunden ab jetzt fest
    NextTick = Now + TimeValue("00:00:05")
    Application.OnTime NextTick, "UpdateClock"
End Sub

Sub StopClock()
'   Bricht das OnTime-Ereignis ab (stoppt die Uhr)
    On Error Resume Next
    Application.OnTime NextTick, "UpdateClock", , False
End Sub





Sub Setup_OnKey()
    Application.OnKey "{PgDn}", "PgDn_Sub"
    Application.OnKey "{PgUp}", "PgUp_Sub"
End Sub

Sub PgDn_Sub()
    On Error Resume Next
      ActiveCell.Offset(1, 0).Activate
End Sub

Sub PgUp_Sub()
    On Error Resume Next
      ActiveCell.Offset(-1, 0).Activate
End Sub





Sub Cancel_OnKey()
    Application.OnKey "{PgDn}"
    Application.OnKey "{PgUp}"
End Sub
