
<customUI xmlns='http://schemas.microsoft.com/office/2006/01/customui'>
<ribbon>
<tabs>
<tab idMso='TabHome'>
  <group  id='Group1' label='Excel VBA For Dummies'>
    <button id='Button1' 
        label='Click Me' 
        size='large'
        onAction='ShowMessage' 
        imageMso='FileStartWorkflow' />
    </group>
</tab>
</tabs>
</ribbon>
</customUI>




Sub ShowMessage(control As IRibbonControl)
    MsgBox "Glückwunsch! Sie haben den neuen Menübandbefehl gefunden."
End Sub





Sub ShowShortcutMenusName()
    Dim Row As Long
    Dim cbar As CommandBar
    Row = 1
    For Each cbar In Application.CommandBars
        If cbar.Type = msoBarTypePopup Then
            Cells(Row, 1) = cbar.Index
            Cells(Row, 2) = cbar.Name
            Row = Row + 1
        End If
    Next cbar
End Sub




Sub ShowCaption()
    MsgBox Application.CommandBars("Cell"). _
       Controls(1).Caption
End Sub





Sub AccessControlByName()
    MsgBox CommandBars("Cell").Controls("Copy").Caption
End Sub

Sub AccessControlById()
    MsgBox CommandBars("Cell").FindControl(ID:=19).Caption
End Sub




Sub ShowShortcutMenuItems()
  Dim Row As Long, Col As Long
  Dim Cbar As CommandBar
  Dim Ctl As CommandBarControl
  Row = 1
  Application.ScreenUpdating = False
  For Each Cbar In Application.CommandBars
    If Cbar.Type = msoBarTypePopup Then
      Cells(Row, 1) = Cbar.Index
      Cells(Row, 2) = Cbar.Name
      Col = 3
      For Each Ctl In Cbar.Controls
        If Ctl.Visible Then
          Cells(Row, Col) = Ctl.Caption
        Else
          Cells(Row, Col) = "<" & Ctl.Caption & ">"
        End If
       Col = Col + 1
     Next Ctl
     Row = Row + 1
   End If
  Next Cbar
End Sub





Sub ResetAll()
    Dim cbar As CommandBar
    For Each cbar In Application.CommandBars
        If cbar.Type = msoBarTypePopup Then
            cbar.Reset
            cbar.Enabled = True
        End If
    Next cbar
End Sub





Sub AddToShortCut()
    Dim Bar As CommandBar
    Dim NewControl As CommandBarButton
    DeleteFromShortcut
    Set Bar = Application.CommandBars("Cell")
    Set NewControl = Bar.Controls.Add _
        (Type:=msoControlButton, ID:=1, _
         temporary:=True)
    With NewControl
        .Caption = "&Change Case"
        .OnAction = "ChangeCase"
        .Style = msoButtonIconAndCaption
    End With
End Sub





Sub DeleteFromShortcut()
    On Error Resume Next
    Application.CommandBars("Cell").Controls _
      ("&Change Case").Delete
End Sub





Private Sub Workbook_Open()
    Call AddToShortCut
End Sub

Private Sub Workbook_BeforeClose(Cancel As Boolean)
    Call DeleteFromShortcut
End Sub





Private Sub Workbook_Open()
    Application.CommandBars("Cell").Enabled = False
End Sub





Private Sub Workbook_BeforeClose(Cancel As Boolean)
    Application.CommandBars("Cell").Enabled = True
End Sub





Sub MakeOldMenus()
   Dim cb As CommandBar
   Dim cbc As CommandBarControl
   Dim OldMenu As CommandBar

'  Löschen, falls bereits vorhanden
   On Error Resume Next
   Application.CommandBars("Old Menus").Delete
   On Error GoTo 0

'  Eine alte Menüleiste erstellen
'  Set the last argument to False for a more compact menu
   Set OldMenu = Application.CommandBars.Add("Old Menus", , True)

'  Die Steuerelemente aus dem Kontextmenü "Built-in Menus" von Excel kopieren
   With CommandBars("Built-in Menus")
     .Controls("&File").Copy OldMenu
     .Controls("&Edit").Copy OldMenu
     .Controls("&View").Copy OldMenu
     .Controls("&Insert").Copy OldMenu
     .Controls("F&ormat").Copy OldMenu
     .Controls("&Tools").Copy OldMenu
     .Controls("&Data").Copy OldMenu
     .Controls("&Window").Copy OldMenu
     .Controls("&Help").Copy OldMenu
   End With

'   Anzeigen auf der Registerkarte Add-Ins
   Application.CommandBars("Old Menus").Visible = True
End Sub
