
Function ChangeComment(cell, NewText)
    cell.Comment.Text NewText
End Function





Function FirstName()
    Dim FullName As String
    Dim FirstSpace As Integer
    FullName = Application.UserName
    FirstSpace = InStr(FullName, " ")
    If FirstSpace = 0 Then
        FirstName = FullName
    Else
        FirstName = Left(FullName, FirstSpace - 1)
    End If
End Function





Function User()
'   Gibt den Namen des aktuellen Benutzers zurück.
    User = Application.UserName
End Function





Function Commission(Sales)
'   Berechnet Verkaufskommissionen
    Const Tier1 As Double = 0.08
    Const Tier2 As Double = 0.105
    Const Tier3 As Double = 0.12
    Const Tier4 As Double = 0.14
    Select Case Sales
        Case 0 To 9999.99: Commission = Sales * Tier1
        Case 10000 To 19999.99: Commission = Sales * Tier2
        Case 20000 To 39999.99: Commission = Sales * Tier3
        Case Is >= 40000: Commission = Sales * Tier4
    End Select

    Commission = Round(Commission, 2)
End Function





Function Commission2(Sales, Years)
'   Berechnet Verkaufskommissionen nach Zugehörigkeitsjahren
    Const Tier1 As Double = 0.08
    Const Tier2 As Double = 0.105
    Const Tier3 As Double = 0.12
    Const Tier4 As Double = 0.14
    Select Case Sales
        Case 0 To 9999.99: Commission2 = Sales * Tier1
        Case 10000 To 19999.99: Commission2 = Sales * Tier2
        Case 20000 To 39999.99: Commission2 = Sales * Tier3
        Case Is >= 40000: Commission2 = Sales * Tier4
    End Select
    Commission2 = Commission2 + (Commission2 * Years / 100)
    Commission2 = Round(Commission2, 2)
End Function





Function JoinText(InRange, Delim)
    Dim Cell As Range
    For Each Cell In InRange
        JoinText = JoinText & Cell.Value & Delim
    Next Cell
    JoinText = Left(JoinText, Len(JoinText) - Len(Delim))
End Function





Function TopAvg(InRange, N)
'   Gibt den Durchschnitt der N höchsten Werte in InRange zurück
    Dim Sum As Double
    Dim i As Long
    Sum = 0
    For i = 1 To N
      Sum = Sum + WorksheetFunction.Large(InRange, i)
    Next i
    TopAvg = Sum / N
End Function





Function DrawOne(InRange, Optional Recalc = 0)
'   Wählt eine zufällige Zelle aus einem Bereich aus

    Randomize
'   Funktion Volatile machen, wenn Recalc gleich 1 ist
    If Recalc = 1 Then Application.Volatile True

'   Zufällige Zelle bestimmen
    DrawOne = InRange(Int((InRange.Count) * Rnd + 1))
End Function





Function User()
'   Gibt den Namen des aktuellen Benutzers zurück
    User = Application.UserName
End Function





Function NumberFormat(Cell)
'   Gibt das Zahlenformat der Zelle zurück
    NumberFormat = Cell(1).NumberFormat
End Function





Function ExtractElement(Txt, n, Sep)
'   Gibt das n-te Element einer Zeichenfolge zurück,
'   deren Elemente durch ein vorgegebenes Trennzeichen
'   getrennt sind
    ExtractElement = Split(Application.Trim(Txt), Sep)(n - 1)
End Function





Function SayIt(txt)
'   Spricht das Argument aus
    Application.Speech.Speak txt, True
End Function





Function IsLike(text, pattern)
'   Gibt True zurück, wenn das erste Argument gleich 
'   dem zweiten ist.
    IsLike = text Like pattern
End Function





Function MonthNames()
    MonthNames = Array("Januar", "Februar", "März", _
      "April", "Mai", "Juni", "Juli", "August", _
      "September", "Oktober", "November", "Dezember")
End Function





Function Sorted(Rng As Range)
    Dim SortedData() As Variant
    Dim Cell As Range
    Dim Temp As Variant, i As Long, j As Long
    Dim NonEmpty As Long
   
'   Daten in SortedData übertragen
    For Each Cell In Rng
        If Not IsEmpty(Cell) Then
            NonEmpty = NonEmpty + 1
            ReDim Preserve SortedData(1 To NonEmpty)
            SortedData(NonEmpty) = Cell.Value
        End If
    Next Cell

'   Das Array sortieren
    For i = 1 To NonEmpty
        For j = i + 1 To NonEmpty
            If SortedData(i) > SortedData(j) Then
                Temp = SortedData(j)
                SortedData(j) = SortedData(i)
                SortedData(i) = Temp
            End If
        Next j
    Next i
        
'   Das Array umformen und zurückgeben
    Sorted = Application.Transpose(SortedData)
End Function




Sub AddArgumentDescriptions()
   Application.MacroOptions Macro:="TopAvg", _
       ArgumentDescriptions:= _
        Array("Bereich, der die Werte enthält", _
        "Anzahl der Werte für den Durchschnitt")
End Sub
