
Sub EnterSquareRoot()
    Dim Num As Double
'   Prompt for a value
    Num = InputBox("Geben Sie einen Wert ein")

'   Quadratwurzel einfügen
    ActiveCell.Value = Sqr(Num)
End Sub





Sub EnterSquareRoot2()
    Dim Num As Double
'   Prompt for a value
    Num = InputBox("Geben Sie einen Wert ein")

'   Sicherstellen, dass die Zahl nicht negativ ist
    If Num < 0 Then
        MsgBox "Sie müssen eine positive Zahl eingeben."
        Exit Sub
    End If

'   Quadratwurzel eintragen
    ActiveCell.Value = Sqr(Num)
End Sub





Sub EnterSquareRoot3()
    Dim Num As Variant
'   Prompt for a value
    Num = InputBox("Geben Sie einen Wert ein")

'   Sicherstellen, dass Num eine Zahl ist
    If Not IsNumeric(Num) Then
        MsgBox "Sie müssen eine Zahl eingeben."
        Exit Sub
    End If

 '   Sicherstellen, dass die Zahl nicht negativ ist
    If Num < 0 Then
        MsgBox "Sie müssen eine positive Zahl eingeben."
        Exit Sub
    End If

'   Insert the square root
    ActiveCell.Value = Sqr(Num)
End Sub





Sub EnterSquareRoot4()
    Dim Num As Variant
'   Sicherstellen, dass ein Arbeitsblatt aktiv ist
    If TypeName(Selection) <> "Range" Then
        MsgBox "Wählen Sie eine Zelle für das Ergebnis aus."
        Exit Sub
    End If

'   Prompt for a value
    Num = InputBox("Geben Sie einen Wert ein")

'   Sicherstellen, dass Num eine Zahl ist
    If Not IsNumeric(Num) Then
        MsgBox "Sie müssen eine Zahl eingeben."
        Exit Sub
    End If

'   Sicherstellen, dass die Zahl nicht negativ ist
    If Num < 0 Then
        MsgBox "Sie müssen eine positive Zahl eingeben."
        Exit Sub
    End If

'   Die Quadratwurzel einfügen
    ActiveCell.Value = Sqr(Num)
End Sub





Sub EnterSquareRoot5()
    Dim Num As Variant
    Dim Msg As String

'   Fehlerverarbeitung einrichten
    On Error GoTo BadEntry

'   Wert abfragen
    Num = InputBox("Geben Sie einen Wert ein")

'   Beenden, falls abgebrochen
    If Num = "" Then Exit Sub

'   Die Quadratwurzel einfügen
    ActiveCell.Value = Sqr(Num)
    Exit Sub

BadEntry:
    Msg = "Ein Fehler ist aufgetreten." & vbNewLine & vbNewLine
    Msg = Msg & "Achten Sie darauf, dass ein Bereich ausgewählt ist, "
    Msg = Msg & "dass das Blatt nicht geschützt ist "
    Msg = Msg & "und Sie einen nicht negativen Wert eingegeben haben."
    MsgBox Msg, vbCritical
End Sub





Sub EnterSquareRoot6()
    Dim Num As Variant
    Dim Msg As String
    Dim Ans As Integer
TryAgain:
'   Einrichtung der Fehlerverarbeitung
    On Error GoTo BadEntry

'   Einen Wert abfragen
    Num = InputBox("Geben Sie einen Wert ein")
    If Num = "" Then Exit Sub

'   Die Quadratwurzel einfügen
    ActiveCell.Value = Sqr(Num)

    Exit Sub

BadEntry:
    Msg = Err.Number & ": " & Error(Err.Number)
    Msg = Msg & vbNewLine & vbNewLine
    Msg = Msg & "Achten Sie darauf, dass ein Bereich ausgewählt ist, "
    Msg = Msg & "dass das Blatt nicht geschützt ist, "
    Msg = Msg & "und dass Sie einen nicht negativen Wert eingeben."
    Msg = Msg & vbNewLine & vbNewLine & "Wiederholen?"
    Ans = MsgBox(Msg, vbYesNo + vbCritical)
    If Ans = vbYes Then Resume TryAgain
End Sub





Sub SelectionSqrt()
    Dim cell As Range
    If TypeName(Selection) <> "Range" Then Exit Sub
    For Each cell In Selection
        cell.Value = Sqr(cell.Value)
    Next cell
End Sub





Sub SelectionSqrt()
    Dim cell As Range
    If TypeName(Selection) <> "Range" Then Exit Sub
    On Error Resume Next
    For Each cell In Selection
        cell.Value = Sqr(cell.Value)
    Next cell
End Sub





Sub SelectionSqrt()
    Dim cell As Range
    Dim ErrMsg As String
    If TypeName(Selection) <> "Range" Then Exit Sub
    On Error GoTo ErrorHandler
    For Each cell In Selection
        cell.Value = Sqr(cell.Value)
    Next cell
    Exit Sub

ErrorHandler:
    Select Case Err.Number
       Case 5 'Negative Zahl
         Resume Next
       Case 13 'Falscher Typ
         Resume Next
       Case 1004 'Zelle gesperrt, Blatt geschützt
         MsgBox "Die Zelle ist gesperrt. Erneut versuchen.", vbCritical, cell.Address
         Exit Sub
       Case Else
         ErrMsg = Error(Err.Number)
         MsgBox "ERROR: " & ErrMsg, vbCritical, cell.Address
         Exit Sub
    End Select
End Sub





Function WorkbookIsOpen(book As String) As Boolean
    Dim WBName As String
    On Error GoTo NotOpen
    WBName = Workbooks(book).Name
    WorkbookIsOpen = True
    Exit Function
NotOpen:
    WorkbookIsOpen = False
End Function





Function WorkbookIsOpen(book) As Boolean
    Dim WBName As String
    On Error Resume Next
    WBName = Workbooks(book).Name
    If Err.Number = 0 Then WorkbookIsOpen = True _
      Else WorkbookIsOpen = False
End Function





Sub UpdatePrices()
    If Not WorkbookIsOpen("Prices.xlsx") Then
        MsgBox "Bitte öffnen Sie zuerst die Arbeitsmappe Prices!"
        Exit Sub
    End If
'   [Hier kann weiterer Code stehen]
End Sub
